```
> restart;
> eq := diff(x(t),t,t) + omega0^2*x(t) = 0;
> sol := dsolve({eq, x(0)=x0, D(x)(0)=v0}, x(t));
> assign(sol);
> x(t);
> E := 1/2*m*diff(x(t),t)^2 + 1/2*k*x(t)^2;
> E := simplify(eval(E, omega0=sqrt(k/m)));
> Ek := 1/2*m*diff(x(t),t)^2;
> Ep := 1/2*k*x(t)^2;
> T0 := 2*Pi/omega0;
> Ek_s := int(Ek, t=0..T0) / T0;
> simplify(eval(Ek_s, omega0=sqrt(k/m)));
> Ep_s := int(Ep, t=0..T0) / T0;
> simplify(eval(Ep_s, omega0=sqrt(k/m)));

> eq1 := A*cos(omega0*t + theta) = x(t);
> eq2 := eval(eq1, t=0);
> eq3 := eval(eq1, t=Pi/(2*omega0));
> solve({eq2,eq3}, {A,theta});
> A := solve(eq2,A);
> theta := solve(eq3,theta);
> A;
> x0 := 'x0'; v0 := 'v0';
> X := unapply(x(t),x0,v0,omega0,t);
> V := unapply(diff(x(t),t),x0,v0,omega0,t);
> plot({[X(1,0,1,t),V(1,0,1,t),t=0..2*Pi],
        [X(2,0,1,t),V(2,0,1,t),t=0..2*Pi],
        [X(3,0,1,t),V(3,0,1,t),t=0..2*Pi]});

Tlumeny harmonicky pohyb
> restart;
> eq := diff(x(t),t,t)+2*beta*diff(x(t),t)+omega0^2*x(t)=0;
> sol := dsolve({eq,x(0)=x0,D(x)(0)=v0},x(t));
> simplify(eval(eq, sol));
> assign(sol);
> x(t);
> X := unapply(x(t), x0, v0, omega0, beta,t);
> plot({X(1,0,1,1/8,t),
        limit(X(1,0,1,beta,t), beta=1),
        X(1,0,1,4,t)}, t=0..6*Pi);
Kriticke tlumeni(β = ω0)
> limit(x(t), omega0 = beta);
```

```
> restart;
> eq := diff(x(t),t,t)+2*beta*diff(x(t),t)
        + omega0^2*x(t)=F0/m*sin(omega*t);
> sol := dsolve(eq,x(t));
> assign(sol);
> simplify(eq);
> x(t):='x(t)';
> sol := dsolve({eq,x(0)=x0,D(x)(0)=v0},x(t));
> assign(collect(%,F0,simplify));
> x(t);
> v1:=eval(x(t), {m=1, F0=1, beta=0.1, omega=1, omega0=1});
> plot([eval(v1,{x0=cos(0),v0=sin(0)}), eval(v1,{x0=cos(Pi/3),v0=
  sin(Pi/3)}), eval(v1,{x0=cos(Pi/4),v0=sin(Pi/4)})],t=0..100);
> v:=diff(x(t),t);
> K:=1/2*m*v^2;
> U:=1/2*m*(omega0*x(t))^2;
> En:=unapply(Re(K+U), t, x0, v0, m, omega0, beta, F0, omega);
> En2 := (t,beta,omega) -> En(t,0,0, 1,1,beta, 1,omega);
> plot({En2(t,1/10,0.5), En2(t,1/10,0.9), En2(t,1/10,1.0)}, t=0.
  .10/(1/10));
> plot3d(En2(t,1/10,omega),t=0..30,omega=0..2,
      grid=[60,40], axes=frame,orientation=[-60,60],view=0..12);
> T:=2*Pi/omega;
> i1 := eval(int(v*F0*sin(omega*t),t=t0..t0+T)/T,
              {x0=0, v0=0, m=1, omega0=1, F0=1});
> AvgPower := unapply(i1, t0,beta,omega):
> P0 := (beta,omega) -> Re(AvgPower(0,beta,omega)):
> Pss := (beta,omega) -> Re(AvgPower(10/beta,beta,omega)):
> plot({P0(0.1,omega),Pss(0.1,omega)},omega=0..4);
> plot({Pss(0.1,omega),Pss(0.25,omega),Pss(0.5,omega)}, omega=0..4)
  ;
```